M 365 Excel Class Video 06: Text Formulas and Functions by excelisfun

Table of Contents

Excel Worksheet Text Functions

Text operators:

& = Ampersand = Join Operator

Any Math Operation on Text Numbers will convert them back to numbers. It will also remove extra spaces.

Text Values is often referred to as **"Text Strings"**.

Text Functions:

TEXTJOIN functions joins text from multiple cells into one cell based on a delimiter.

CONCAT joins text from multiple cells into one cell.

TEXTAFTER function extracts text after a delimiter

TEXTBEFORE function extracts text before a delimiter

LEN function counts characters

TEXTSPLIT will split text by delimiters

TEXT function converts a number to text with a Custom Number Format that you specify, like: TEXT(number, "Custom Number Format in quotes")

FIXED function converts a number to text based on Number Format

DOLLAR function converts a number to text based on Currency Number Format

LEFT extracts a given number of characters from the left

RIGHT extracts a given number of characters from the right

ADDRESS delivers the cell address for a specified cell

FORMULATEXT delivers a formula as text from a specified cell

SEARCH tells you the starting position in a text string of text you specify. **FIND** is similar to SEARCH, but it is case sensitive.

REPLACE function replaces part of a text string with text you specify, given a starting number and the number of characters

SUBSTITUTE function finds some text and replaces it with some different text

TRIM removes spaces from a text string except for single spaces between words

MID extracts from the middle given a starting point and the number of characters that you want

PROPER function changes all capital letters or all lower case letters to all lower case except for first letter in each word

LOWER converts all letters to lower case. **UPPER converts** all letters to upper case

CODE function gives you numeric code given a character. **CHAR function** gives you character given a numeric code.

Worksheet Formula Video Examples:



Join Text

First Name	Last Name	Join
Rosalie	Mims	Rosalie Mims
Kenny	Noline	Kenny Noline
Kenny	Noline	KennyNoline

D32: =B32&" "&C32
D33: =TEXTJOIN(" ",,B33:C33)
D34: =CONCAT(B34:C34)

Extract Text, then Lookup

Product ID	Name	Price
CAR-4369-01	Carlota	28.95
QUA-0101	Quad	43.95

C39: =XLOOKUP(TEXTBEFORE(TEXTAFTER(B39,"-"),"-")+0,B45:B46,C45:D46)
C40: =XLOOKUP(TEXTAFTER(B40,"-")+0,B45:B46,C45:D46)

Lookup Table:

Product ID	Name	Price
4369	Carlota	28.95
101	Quad	43.95

Length of Text String?

Length?	4	Lookup	Quad
Length?	4	Price	43.95

F51: =XLOOKUP(F50,C45:C46,D45:D46)

Split Text

Description	Product	Amount	Region
Carlota/658	Carlota	658	
Aspen/345:South	Aspen	345	South

Description	Carlota	658
Carlota/658:West	West	

Text Formulas to Create Labels that Include Formatted Numbers.

Number	Label
25.754	The amount due is: $25.75
3451.987	The amount due is: 3,451.99
2:00 PM	Your appointment time is: 2:00 PM
12/3/23	The loan is due on: 12/3/2023
9.98%	The rate on the loan is: 9.98%

C62: ="The amount due is: "&DOLLAR(B62)
C63: ="The amount due is: "&FIXED(B63)
C64: ="Your appointment time is: "&TEXT(B64,"h:mm AM/PM")
C65: ="The loan is due on: "&TEXT(B65,"m/d/yyy")
C66: ="The rate on the loan is: "&TEXT(B66,"0.00%")

Extract a fixed number of characters

State-Zip	State	Zip
CA-94704	CA	94704
WA-98106	WA	98106

Show Formulas vertically

=LEFT(B71:B72,2)
=RIGHT(B71:B72,5)

F72: =TRANSPOSE(FORMULATEXT(C71:D71))



Partial Text Record Lookup

E-mail	Name
fun@sc.edu	Finny
neat@gmail.com	Gigi
q@yahoo.com	Quan
mm@sc.edu	Chantel

Criteria: com

E-mail	Name
neat@gmail.com	Gigi
q@yahoo.com	Quan

<= E80: =FILTER(B78:C81,ISNUMBER(SEARCH(F77,B78:B81)))

E-mail	Name
neat@gmail.com	Gigi
q@yahoo.com	Quan

<= H83: =FILTER(B78:C81,RIGHT(B78:B81,3)=F77)

Replace, Insert and Substitute text into a text string

Data	AAAAA-DD
FVVRW00091	FVVRW-91
WTKRY00052	WTKRY-52

C87: =REPLACE(B87:B88,6,3,"-")

Data	AAAAA-00000
FVVRW00091	FVVRW-00091
WTKRY00052	WTKRY-00052

C91: =REPLACE(B91:B92,6,0,"-")

Data	AAAAA-GGG
FVVRW00091	FVVRWGGG91
WTKRY00052	WTKRYGGG52

C95: =SUBSTITUTE(B95:B96,"000","GGG")

Remove all spaces except single spaces between words

Web Site	TRIM	Len	
Amazon Store	Amazon Store	15	12
Gel-boomerangs	Gel-boomerangs	14	14
E-bay	E-bay	6	5
Amazon.com	Amazon.com	11	10
Amazon.com	Amazon.com	12	10
E-bay	E-bay	9	5
Gel-boomerangs	Gel-boomerangs	14	14
Amazon Prime	Amazon Prime	18	12
E-bay	E-bay	5	5
Amazon	Amazon	8	6

C101: =TRIM(B101:B110)

E101: =LEN(B101:C110)

Remove all spaces & Character 160 except single spaces between words

Web Site	TRIM	Len	
Amazon Store	Amazon Store	18	12
Gel-boomerangs	Gel-boomerangs	14	14
E-bay	E-bay	6	5
Amazon.com	Amazon.com	11	10
Amazon.com	Amazon.com	12	10
E-bay	E-bay	7	5
Gel-boomerangs	Gel-boomerangs	14	14
Amazon Prime	Amazon Prime	18	12
E-bay	E-bay	5	5
Amazon	Amazon	8	6

C115: =TRIM(SUBSTITUTE(B115:B124,CHAR(160)," "))

E115: =LEN(B115:C124)

Analy

Web Site
Amazon Store



Analyze text

160 = No-breaking Space

32 = Spacebar space

Web Site
Amazon Store

Character	Code
	160
	160
A	65
m	109
a	97
z	122
o	111
n	110
	32
	32
	32
S	83
t	116
o	111
r	114
e	101
	32
	32

24,CHAR(160)," "))

(C131:C135),"",C131:C13

L115: =MID(J114,SEQUENCE(LEN(J114)),1)

M115: =CODE(L115#)



	A	B	C	D	E	F	G	H	I	J	K
127											
128		**TEXTJOIN!**									
129											
130		First	Middle	Last	TEXTJOIN						
131		Sioux		Radcoolinator	Sioux Radcoolinator	Sioux Radcoolinator		E131: =TEXTJOIN(" ",,B131:D131)			
132		Gigi	T.	Fran	Gigi T. Fran	Gigi T. Fran		F131: =B131:B135&" "&IF(ISBLANK(C131:C135),"",C131:C135)			
133		Tyrone	Chip	Smith	Tyrone Chip Smith	Tyrone Chip Smith					
134		Chin		Pham	Chin Pham	Chin Pham					
135		Dennis	Big D	Ho	Dennis Big D Ho	Dennis Big D Ho					
136											
137		E-mails			TEXTJOIN						
138		MamieH@gnet.com			MamieH@gnet.com;DebiD@yahoo.com;PalmerA@yahoo.com;ZinaW@gmail.com;SanfordB@hotmail.com;HildeC@gnet.com;P						
139		DebiD@yahoo.com									
140		PalmerA@yahoo.com			E138: =TEXTJOIN(";",,B138:B157)						
141		ZinaW@gmail.com									

Power Query Video Examples:

External File that you need to import: Power Query

```
State-Zip → Sales·Rep → Description›Web·Site¶
CA-94704 → Rosalie·Mullins → Carlota·/·West·/·658 → °°Amazon···Store··¶
WA-98106 → Marcia·Parker → Aspen·/·South·/·345 → Gel-boomerangs¶
OR-96011 → Christy·Hogan → Yanaki·/·South·/·19.5→E-bay·¶
CA-98702 → Sophia·Maxwell → FlatTop·/·South·/·987.75 → Amazon.com·¶
WA-98108 → Salvador·Craig → Carlota·/·South·/·56 → ·Amazon.com·¶
WA-98112 → Wanda·Stevens → Carlota·/·East·/·23.5→°°E-bay¶
OR-96011 → Harvey·Tucker → Carlota·/·West·/·321 → Gel-boomerangs¶
CA-94600 → Jesse·Kelley → Yanaki·/·East·/·1209.5→Amazon·····Prime···¶
CA-94688 → Miguel·Simpson → Yanaki·/·East·/·123.5→E-bay¶
WA-98342 → Darryl·May→Quad·/·West·/·399.95 → Amazon·····C¶
```


PQTextFunctionsClean-Solution - Power Query Editor
File
Home
Transform
Add Column
View
Close & Load
Refresh Preview
Properties
Advanced Editor
Manage
Close
Query
Choose Columns
Remove Columns
Manage Columns
Keep Rows
Remove Rows
Reduce Rows
Sort
Split Column
Group By
Data Type: Text
Use First Row as Headers
Replace Values
Transform
Merge Queries
Append Queries
Combine Files
Combine
Manage Parameters
Parameters
Data source settings
Data Sources
New Source
Recent Sources
Enter Data
New Query
Queries
= Table.TransformColumnTypes(RemoveInteriorDoubleSpace,{{"State", type text}, {"Zip", type text}, {"FirstName", type text},
State
Zip
FirstName
LastName
Product
Region
Valid 100%
Error 0%
Empty 0%
1 CA 94704 Rosalie Mullins Carlota West
2 WA 98106 Marcia Parker Aspen South
3 OR 96011 Christy Hogan Yanaki South
4 CA 98702 Sophia Maxwell FlatTop South
5 WA 98108 Salvador Craig Carlota South
6 WA 98112 Wanda Stevens Carlota East
7 OR 96011 Harvey Tucker Carlota West
8 CA 94600 Jesse Kelley Yanaki East
9 CA 94688 Miguel Simpson Yanaki East
10 WA 98342 Darryl May Quad West
Query Settings
PROPERTIES
Name
PQTextFunctionsClean-Solution
All Properties
APPLIED STEPS
Source
PromotedHeaders
SplitStateZip
SplitFirstLastName
SplitPRA
Trim
RemoveInteriorDoubleSpace
ChangeDataTypes